
Zusammenfassung 

SAS speichert intern die Wahrheitswerte TRUE und FALSE als numerische Variable mit 
den Ausprägungen 1 bzw. 0 ab. Das kann man sich zunutze machen, um Code zu vereinfa-
chen. 
Das scheinbar sinnlose Konstrukt  IF 0 THEN SET ...  kann genutzt werden, um Variablen 
zu definieren, während ich an den Observations nicht das geringste Interesse habe. 

 
Schlüsselwörter: Datenschritt, Boolesche Variablen, Kompilierungsphase, Ausfüh-
rungsphase 

1 Rechnen mit der Wahrheit 
SAS kennt keine Booleschen Variablen, also Variablen, die die Wahrheitswerte TRUE 
oder FALSE darstellen. Die „Wahrheit“ wird über Zahlen definiert, wobei gilt: 

 Missing values (auch spezielle fehlende Werte wie .A) und 0 sind ‘falsch’ 
(FALSE)  

 Alle anderen Zahlen sind ‘wahr’ (TRUE). 

Ergebnisse von Vergleichsoperationen werden als 1 (TRUE) bzw. 0 (FALSE) ausgege-
ben bzw. zwischengespeichert. Einige Beispiele sollen verdeutlichen, welche Auswir-
kungen dieses Verhalten hat, zumal es nicht immer intuitiv ist. 
 
Der Ausdruck 0.1 < 0.2 < 0.3 wird in der Datenschritt-Programmierung als TRUE 
erkannt. Das entspricht der Intuition. Allerdings wird dieser Ausdruck vor der Bewer-
tung intern umgewandelt in  (0.1 < 0.2) and (0.2 < 0.3). 
 
Der gleiche Ausdruck ist in der Makrosprache FALSE, da diese Umwandlung nicht 
stattfindet und folgendermaßen schrittweise von links abgearbeitet wird: 

            0.1 < 0.2 < 0.3 
 
                1     < 0.3 
 
                     0  = FALSE 

Einige andere, ähnliche Beispiele mit teilweise überraschendem Ergebnis: 
 
       (0.1 < 0.2) < 0.3      => FALSE 
        0.1 < (0.2 < 0.3)     => TRUE 
       (1 < 2) < 3            => TRUE 
 
Des Weiteren macht SAS keinen Unterschied zwischen dem Zuweisungsoperator und 
dem Gleichheitsoperator, beides wird durch das Gleichheitszeichen dargestellt. 
Folgende seltsam anmutende Konstruktion ist daher möglich und syntaktisch völlig 
einwandfrei: 

a = 1 = 2; 

Zuerst wird der Vergleich angestellt: 1 = 2 ist falsch, deshalb ergibt dieser Ausdruck 0. 
Diese 0 wird a zugewiesen. Die Variable a hat also am Ende den Wert 0! 
 
Dieses Verhalten kann man sich zunutze machen, um z. B. IF-Anweisungen anders dar-
zustellen. 

     if age > 50 then age_group = 2; 
                 else age_group = 1; 

kann geschrieben werden als 

     age_group = 2 * (age > 50) + 1 * (age <= 50); 

Von der Richtigkeit dieser Konstruktion überzeugt man sich am besten am Beispiel: 

     Age=60   =>   age_group= 2 * 1 + 1 * 0 = 2 
     Age=40   =>   age_group= 2 * 0 + 1 * 1 = 1 

So kann man Programme kürzer und mit weniger Anweisungen schreiben. Allerdings 
ist diese Art der Programmierung nicht selbsterklärend und sollte unbedingt kommen-
tiert werden. 

1.1 Beispiel: Berechnung des Alters 

Das Alter eines Menschen in Jahren berechnet man intuitiv, indem man das derzeitige 
Jahr von Geburtsjahr abzieht. Danach überprüft man, ob derjenige in diesem Jahr schon 
Geburtstag hatte – wenn ja, wird noch 1 abgezogen. 
SAS bietet leider keine Funktion, die das Alter exakt berechnet. Auch die Funktion 
YRDIF gibt nicht immer das richtige Resultat, siehe http://support.sas.com/ 
/kb/3/036.html. 

Mit ‚konventionellem’ SAS Code würde man programmieren: 

BDate = '05Jun1967 'd; 
Today = today(); 
 
Age = year(Today) - year(BDate); 
 
*** 1 abziehen, wenn in diesem Jahr noch kein Geburtstag war **; 
 
if  month(Today) < month(BDate) or  
    month(Today) = month(BDate) and day(Today) < day(BDate) 
 
    then Age = Age - 1; 

Dies hat den Nachteil, dass es, eingebaut in ein Makro, nicht wie eine einfache Funktion 
verwendet werden kann. Wandelt man die IF-Bedingung um und berechnet das Alter 
mit 

Age = year(Today) - year(BDate) -  
      (month(Today) < month(BDate) or  
       month(Today) = month(BDate) and day(Today) < day(BDate) 
      ); 

dann kann das Makro in einer einfachen Zuweisung benutzt werden, also: 
Age = %Age(BDate, Today) 

1.2 Beispiel: Gruppenmerkmale 

Häufig benötigt man in der Statistik Kennzahlen, die pro Gruppe berechnet werden 
müssen, z. B. für Mittelwertsdifferenzen zweier Gruppen (hier die rote und blaue 
Gruppe): 

MeanDiffblau-rot  =  Summeblau / Anzahlblau – Summerot / Anzahlrot 

Die Datenstruktur ist dabei oft vertikal, d. h. die Gruppenzugehörigkeit ist in einer eige-
nen Variablen gespeichert (siehe Datensatz VALUES in Abbildung 1). 
Mit der Prozedur SQL und der 0-1-Logik gelingt es nun in einem Schritt, die Anzahlen 
und Summen in eine Beobachtung zu bringen: 

proc sql; 
  create table sum as 
  select  
      sum(group='rot' )          as n_rot 
     ,sum(group='blau')          as n_blau 
     ,sum(value*(group='rot' ))  as sum_rot 
     ,sum(value*(group='blau'))  as sum_blau 
  from values 
  ; 

Die Screenshots verdeutlichen diese ‚interne’ Entwicklung des Ausgabedatensatzes: 
 
 
 
 
 
 
 
 
 
 
 

 
 
 
                                                
Abbildung 1: Datensatz VALUES und Ausgabendatensatz SUM 

1.3 Beispiel: Gruppeneigenschaften 

Aufgabe sei es, diejenigen Beobachtungen aller Gruppen zu behalten, bei denen we-
nigstens eine Beobachtung eine bestimmte Eigenschaft hat. 
 
Zum Beispiel sollen beim Datensatz VALUES (s. Abb. 1) alle Gruppen behalten wer-
den, bei denen die Variable VALUE in irgendeiner Beobachtung größer als 2 ist. Das ist 
bei der roten Gruppe der Fall. 
 
proc sql; 
   create table values2 as 
   select *  
   from values 
   group by group 
   having max(value > 2.0); 
 
 

2 IF 0 THEN SET: Der Unterschied zwischen Kompilierung 
und Ausführung 

In einigen Vorträgen auf der KSFE, z. B. [1], wurde bereits auf den Unterschied zwi-
schen „Compilation“ (Kompilierungsphase) und „Execution“ (Ausführungsphase) ein-
gegangen. Dieses Kapitel zeigt ein Beispiel, wie man sich diesen Unterschied zunutze 
machen kann. 

In der Kompilierungsphase wird der Program Data Vector (PDV) erzeugt. Hierzu wer-
den, vereinfacht gesagt, alle Variablen erzeugt und ihre Eigenschaften wie Name, Da-
tentyp, Länge und Label festgelegt in der Reihenfolge ihres Bekanntwerdens im Da-
tastep. In der Ausführungsphase werden die Daten der Eingabedatensätze gelesen und 
alle ausführbaren Anweisungen abgearbeitet. 
 
Die Aufgabe, deren Lösung im Folgenden gezeigt wird, besteht darin, Metadaten eines 
Datensatzes zu benutzen. Die Beobachtungen interessieren nicht. Die eigentlichen Da-
ten werden aus anderen Quellen gelesen. 
 
Der Datensatz VARIABLEN soll für die kommenden Beispiele die Metadaten enthal-
ten: 

data Variablen; 
    length brthdat $8; 
    label brthdat = 'Date of Birth'; 
    ...  
run; 

2.1 Einlesen der Daten aus einem zweiten Datensatz 

Zunächst soll der Fall betrachtet werden, wenn die Daten bereits in einem Datensatz 
vorhanden sind: 
data Werte; 
    x = 8; y =  7; output; 
    x = 1; y = 12; output; 
run; 
 
 
1. Idee (schon ziemlich gut) 
 
data Neu; 
 
    set Variablen(where=(0)) 
        Werte; 
 
run; 
 
NOTE: There were 0 observations read from the data set WORK. 
VARIABLEN. 
      WHERE 0 /* an obviously FALSE WHERE clause */ ; 
NOTE: There were 2 observations read from the data set WORK.WERTE. 
NOTE: The data set WORK.NEU has 2 observations and 3 variables. 

Beide Datensätze werden während der Ausführungsphase gelesen, und nur die Obser-
vations des zweiten Datensatzes werden in den Ausgabedatensatz geschrieben. Damit 
ist die Aufgabe erfüllt. Noch eleganter ist es, wenn der Datensatz VARIABLEN nur 
während der Kompilierungsphase genutzt wird, um den PDV zu erstellen, und während 
der Ausführungsphase ignoriert wird: 
 
2. Idee (noch besser): 
 
data Neu; 
 
    if 0 then set Variablen; 
    set Werte; 
 
run; 
 
NOTE: There were 2 observations read from the data set WORK.WERTE. 
NOTE: The data set WORK.NEU has 2 observations and 3 variables. 
 
Im Log kann man erkennen, dass der Datensatz VARIABLEN während der Ausfüh-
rungsphase nicht gelesen wird. 

2.2 Einlesen der Daten im Datastep 

Nun soll der Fall betrachtet werden, wenn die Beobachtungen innerhalb eines neuen 
Datenschritts erzeugt werden, z. B. mittels einer Schleife: 
 
data Werte; 
    ... 
    do brthdat = '01Jan1960'd to '31Dec2020'd; 
       output; 
    end; 
run; 
 
1. Idee (die schon ziemlich gute Idee von oben ist jetzt nicht mehr gut): 
 
data Werte; 
    set Variablen(where=(0)); 
 
    do brthdat = '01Jan1960'd to '31Dec2020'd; 
         output; 
    end; 
 
run; 

NOTE: There were 0 observations read from the data set WORK. 
VARIABLEN. 
      WHERE 0 /* an obviously FALSE WHERE clause */ ; 
NOTE: The data set WORK.WERTE has 0 observations and 1 variables. 
 
Der Ausgabedatensatz enthält keine Beobachtungen! Der Datenschritt wird nicht ausge-
führt, weil SET keine Beobachtungen liefert. 
 
2. Idee (wieder besser): 
 
data Werte; 
    if 0 then set Variablen; 
 
    do brthdat = '01Jan1960'd to '31Dec2020'd; 
         output; 
    end; 
 
run; 
 
NOTE: The data set WORK.WERTE has 22281 observations and 1 
variables. 
 
Das ist die erwartete Anzahl der Beobachtungen, und die Variable BRTHDAT hat die 
Attribute aus dem Datensatz VARIABLEN geerbt. 

